The P&M Social Network

March 2017

Data import

In[1]:=
```
dataset =
 SemanticImport["/Users/cwoodard/GitHub/PaM2017/graphs/pm-teams-mar2017.csv"]
```

Out[1]=

Node	Name	Exp0	Exp1	Exp2	Exp3
1	Daniel	127	222	313	427
2	Diego	132	216	334	435
3	Chris	124	214	323	422
4	Rob	112	232	315	431
5	Allison	111	217	332	423
6	Kristen	111	231	317	435
7	Ilya	116	214	336	415
8	Isa	115	217	332	434
9	Remy	117	234	335	426
10	Matt	116	224	326	417
11	Aurora	122	232	317	412
12	Ray	136	225	316	435
13	AlexCh	125	236	324	411
14	Colvin	114	227	314	433
15	Vivien	123	234	334	424
16	Kyle	121	223	325	436
17	AlexCo	137	213	335	413
18	EvanC	116	214	333	413
19	Zach	135	222	327	416
20	Will	134	212	313	421

showing 1–20 of 83

Graph construction

First, let's group the data by teams for each experiment:

In[2]:=
```
exp0 = dataset[GroupBy["Exp0"]]
```

Out[2]=

	Node	Name	Exp0	Exp1	Exp2	Exp3
127	1	Daniel	127	222	313	427
	29	Minju	127	222	324	432
	4 total ›					
132	2	Diego	132	216	334	435
	30	Hannah	132	213	315	421
	4 total ›					
124	3	Chris	124	214	323	422
	23	Pranay	124	236	321	434
	4 total ›					
112	4	Rob	112	232	315	431
	39	Benjamin	112	224	311	413
	4 total ›					
111	5	Allison	111	217	332	423
	6	Kristen	111	231	317	435
	4 total ›					
116	7	Ilya	116	214	336	415
	10	Matt	116	224	326	417
	4 total ›					
115	8	Isa	115	217	332	434
	25	Lacie	115	212	315	421
	4 total ›					
117	9	Remy	117	234	335	426
	26	Yichen	117	234	327	436
	4 total ›					
122	11	Aurora	122	232	317	412
	24	SamE	122	216	316	431
	4 total ›					
136	12	Ray	136	225	316	435
	33	Ava	136	233	311	424
	4 total ›					
showing 1–10 of 21						

In[3]:=
```
exp1 = dataset[GroupBy["Exp1"]];
exp2 = dataset[GroupBy["Exp2"]];
exp3 = dataset[GroupBy["Exp3"]];
```

Now let's extract the node IDs:

In[6]:=
```
exp0Nodes = Normal@Values[exp0[All, All, "Node"]]
```

Out[6]=
```
{{1, 29, 52, 63}, {2, 30, 32, 38}, {3, 23, 34, 48}, {4, 39, 54, 83}, {5, 6, 77, 82},
 {7, 10, 18, 71}, {8, 25, 27, 62}, {9, 26, 42, 46}, {11, 24, 59, 78},
 {12, 33, 36, 51}, {13, 40, 47, 69}, {14, 37, 74, 80}, {15, 45, 66, 79},
 {16, 35, 64, 72}, {17, 28, 44}, {19, 43, 49, 57}, {20, 41, 53, 81},
 {21, 50, 65, 76}, {22, 55, 58, 73}, {31, 56, 60, 75}, {61, 67, 68, 70}}
```

In[7]:=
```
exp1Nodes = Normal@Values[exp1[All, All, "Node"]];
exp2Nodes = Normal@Values[exp2[All, All, "Node"]];
exp3Nodes = Normal@Values[exp3[All, All, "Node"]];
```

Okay, now we need some functions or things are going to get really messy:

In[10]:=
```
completeGraphRawPairs[n_] := Flatten[Table[{i, j + 1}, {i, 1, n - 1}, {j, i, n - 1}], 1]
completeGraphMappedPairs[nodes_List] :=
 Part[nodes, #] & /@ completeGraphRawPairs[Length[nodes]]
completeGraphEdges[nodes_List] :=
 UndirectedEdge[#[[1]], #[[2]]] & /@ completeGraphMappedPairs[nodes]
```

Voilà! Our first graph:

In[13]:=
```
graph0 = Graph@Flatten[completeGraphEdges /@ exp0Nodes]
```

Out[13]=



In[14]:=
```
graph1 = Graph@Flatten[completeGraphEdges /@ exp1Nodes];
graph2 = Graph@Flatten[completeGraphEdges /@ exp2Nodes];
graph3 = Graph@Flatten[completeGraphEdges /@ exp3Nodes];
```

Finally, we string them together:

In[17]:=
```
graph01 = GraphUnion[graph0, graph1]
```

Out[17]=



In[18]:=
```
graph012 = GraphUnion[graph01, graph2]
```

Out[18]=



In[19]:=
```
graph0123 = GraphUnion[graph012, graph3]
```

Out[19]=



And here's an animation, just for fun:

In[20]:=
```
gl0 = Labeled[graph0, "After Experiment 0"];
gl01 = Labeled[graph01, "After Experiment 1"];
gl012 = Labeled[graph012, "After Experiment 2"];
gl0123 = Labeled[graph0123, "After Experiment 3"];
```

In[24]:=
```
graphSequence = {graph0, graph01, graph012, graph0123};
labeledGraphSequence = {gl0, gl01, gl012, gl0123};
```

In[26]:=
```
anim = Animate[Part[labeledGraphSequence, i],
  {i, 1, 4, 1}, SaveDefinitions → True, AnimationRunning → False]
```

Out[26]=



Which we can deploy to the cloud …

In[27]:=
```
CloudDeploy[anim]
```

Out[27]=
```
CloudObject[
 https://www.wolframcloud.com/objects/c6c235a2-08ac-421d-811c-6e0a69a93870]
```

Graph analysis

And here is some rudimentary graph analysis:

```
In[28]:= ConnectedGraphQ[graph0]
Out[28]= False

In[29]:= ConnectedGraphQ[graph01]
Out[29]= True

In[30]:= VertexCount /@ graphSequence
Out[30]= {83, 83, 83, 83}

In[31]:= meanGraphDistances = MeanGraphDistance /@ graphSequence // N
Out[31]= {∞, 3.12195, 2.39935, 2.11284}

In[32]:= Max@GraphDistanceMatrix[#] & /@ graphSequence
Out[32]= {∞, 6, 4, 3}
```

So — successive randomizing makes the graph smaller … no surprise there!

Now your turn …

I'm out of time, but there's lots of other fun stuff to do here:

• Are there any especially interesting nodes, from a network point of view? (We wouldn't expect any "Kevin Bacons" in a random graph, but someone has to be the most central node …)

• Can you compute statistics on people staying in the same studio between experiments? (How many people have been in the same studio *n* times in a row? Has everyone now been in the same studio with every other member of the Class of 2020 at least once?)

• What happens if you plot relationships between teams instead of individuals? (Could you lay out the graph from left to right, using directed edges to show people moving from one team to the next? Can you track the progression of project topics through the three main experiments?)

Enjoy!